


NCBO SPARQL Endpoint

Trish Whetzel
Outreach Coordinator



THE NATIONAL CENTER FOR
BIOMEDICAL ONTOLOGY

```

1 PREFIX omv: <http://omv.ontoware.org/2005/05/ontology
2
3 SELECT ?ont ?name ?acr
4 WHERE {
5     ?ont a omv:Ontology .
6     ?ont omv:acronym ?acr .
7     ?ont omv:name ?name .
8 }
9

```

Results:


BioPortal SPARQL is a service to query BioMedical ontologies using the SPARQL standard. Ontologies have been transformed into RDF triples from their original formats (OWL, OBO and UMLS/RRF, ...) and asserted into a triple store. This service provides programatic access to that triple store.

[Documentation](#)

[SPARQL Examples](#)


ont	name	acr
<http://bioportal.bioontology.org/ontologies/44840> ↗	"Human disease"	"DOID"
<http://bioportal.bioontology.org/ontologies/46187> ↗	"HOM-DATASOURCE_OSHPD"	"HOM-DATASOURCE_OSHPD"
<http://bioportal.bioontology.org/ontologies/46255> ↗	"Common Anatomy Reference Ontology"	"CARO"
<http://bioportal.bioontology.org/ontologies/40477> ↗	"NanoParticle Ontology"	"NPO"
<http://bioportal.bioontology.org/ontologies/42598> ↗	"Gene Ontology"	"GO"
<http://bioportal.bioontology.org/ontologies/45037> ↗	"Yeast phenotypes"	"YPO"
<http://bioportal.bioontology.org/ontologies/40654> ↗	"NanoParticle Ontology"	"NPO"
<http://bioportal.bioontology.org/ontologies/46097> ↗	"Mosquito insecticide resistance"	"MIRO"
<http://bioportal.bioontology.org/ontologies/42069> ↗	"Cereal plant trait"	"TO"
<http://bioportal.bioontology.org/ontologies/47271> ↗	"HOM-PCSTEST"	"HOM-PCSTEST"
<http://bioportal.bioontology.org/ontologies/45990> ↗	"HOM-ICD9CM-ECODES"	"HOM-ICD9CM-ECODES"

<http://sparql.bioontology.org/>




BioPortal SPARQL Content

- All ontologies from BioPortal
 - Original ontology format (OBO, OWL, UMLS/RRF) transformed into RDF
 - Updated daily
 - *Latest version* only
- Statistics
 - 393 ontologies
 - 4.2M terms
 - 2419 different predicates
 - 80M triples



Resource Description Framework

- A collection of RDF statements represents a labeled, directed multi-graph
- RDF is modeled as a set of triples
- Triples are in the form (subject, predicate, object) where:
 - Subjects and predicates are URIs
 - Objects are either URIs or Literals
 - Literals can be typed and have a language tag
- Blank nodes (bnodes) or anonymous resources are nodes that do not have a URI or literal
- RDF is not XML, there are different serializations, e.g. N3, turtle, rdf/xml




RDF – Turtle format

```
@prefix rdf: <http://www.w3.org/1999/02/22-rdf-syntax-ns#> .
@prefix rdfs: <http://www.w3.org/2000/01/rdf-schema#> .
@prefix owl: <http://www.w3.org/2002/07/owl#> .
@prefix xsd: <http://www.w3.org/2001/XMLSchema#> .

<http://who.int/bodysystem.owl#NervousSystem>
  rdf:type owl:Class ;
  rdfs:label "Nervous System"^^xsd:string ;
  rdfs:subClassOf <http://who.int/bodysystem.owl#BodySystem> .

<http://who.int/bodysystem.owl#BodySystem>
  rdf:type owl:Class ;
  rdfs:label "Body System"^^xsd:string .
```



RDF – N3 format


```
#(each triple block has to be in one line, splitted to fit in sc
<http://who.int/bodysystem.owl#BodySystem>
<http://www.w3.org/1999/02/22-rdf-syntax-ns#type>
<http://www.w3.org/2002/07/owl#Class> .

<http://who.int/bodysystem.owl#BodySystem>
<http://www.w3.org/2000/01/rdf-schema#label>
"Body System"^^<http://www.w3.org/2001/XMLSchema#string> .

<http://who.int/bodysystem.owl#NervousSystem>
<http://www.w3.org/1999/02/22-rdf-syntax-ns#type>
<http://www.w3.org/2002/07/owl#Class> .


<http://who.int/bodysystem.owl#NervousSystem>
<http://www.w3.org/2000/01/rdf-schema#label>
"Nervous System"^^<http://www.w3.org/2001/XMLSchema#string> .

<http://who.int/bodysystem.owl#NervousSystem>
<http://www.w3.org/2000/01/rdf-schema#subClassOf>
<http://who.int/bodysystem.owl#BodySystem> .
```



RDF – RDF/XML format


```
<?xml version="1.0" encoding="utf-8"?>
<rdf:RDF xmlns:rdf="http://www.w3.org/1999/02/22-rdf-syntax-ns#">
  <ns1:Class xmlns:ns1="http://www.w3.org/2002/07/owl#"
    rdf:about="http://who.int/bodysystem.owl#BodySystem">
    <ns2:label xmlns:ns2="http://www.w3.org/2000/01/rdf-schema#"
      rdf:datatype="http://www.w3.org/2001/XMLSchema#string">Body System</ns2:label>
    </ns1:Class>
  <ns3:Class xmlns:ns3="http://www.w3.org/2002/07/owl#"
    rdf:about="http://who.int/bodysystem.owl#NervousSystem">
    <ns4:label xmlns:ns4="http://www.w3.org/2000/01/rdf-schema#"
      rdf:datatype="http://www.w3.org/2001/XMLSchema#string">Nervous System</ns4:label>
    <ns5:subClassOf xmlns:ns5="http://www.w3.org/2000/01/rdf-schema#"
      rdf:resource="http://who.int/bodysystem.owl#BodySystem" />
    </ns3:Class>
</rdf:RDF>
```



SPARQL


- W3C standard query language for RDF
- Structure

```
# prefixes, optional
PREFIX owl: <http://www.w3.org/2002/07/owl#>
...
# result clause {SELECT, DESCRIBE, CONSTRUCT, ASK}
SELECT ...
# dataset definition, optional
FROM ...
# query pattern(s)
WHERE {
    ...
}
# query modifiers, optional
ORDER BY ...
```

BioPortal Metadata

- Virtual ontology identifier
 - Stable identifier across all versions of the ontology
 - All versions of an ontology are linked via this ID
- Ontology version identifier
 - Unique for each ontology version
 - Most metadata linked directly to the ontology version



Virtual Ontology



meta:hasVersion

Versions



name

date

format

(...)

ontology /46116


ontology /46116

```
PREFIX meta: <http://bioportal.bioontology.org/metadata/def/>
SELECT DISTINCT ?graph
WHERE {
  [] meta:hasDataGraph ?graph
}
ORDER BY ?graph
```

graph

http://bioportal.bioontology.org/ontologies/3.0_OWL 


<http://bioportal.bioontology.org/ontologies/AAO> 

<http://bioportal.bioontology.org/ontologies/ABA> 


<http://bioportal.bioontology.org/ontologies/ACGT> 


<http://bioportal.bioontology.org/ontologies/ADW> 

http://bioportal.bioontology.org/ontologies/AEO_x1 

http://bioportal.bioontology.org/ontologies/AERO_x1 

<http://bioportal.bioontology.org/ontologies/AIR> 

<http://bioportal.bioontology.org/ontologies/AMINO-ACID> 

http://bioportal.bioontology.org/ontologies/AMINO_ACID-ZH_CN 

http://bioportal.bioontology.org/ontologies/AMINO_ACID 

Select ontology abbreviations

```

@prefix rdf: <http://www.w3.org/1999/02/22-rdf-syntax-ns#>
@prefix xsd: <http://www.w3.org/2001/XMLSchema#>
@prefix foaf: <http://xmlns.foaf.org/2000/01/01/foaf#>
@prefix rdfs: <http://www.w3.org/2000/01/01/rdf-schema#>
@prefix dc: <http://purl.org/dc/terms/>
@prefix dct: <http://purl.org/dc/dcterm/1.1/>
@prefix owl: <http://www.w3.org/2002/07/owl#>
@prefix bio: <http://purl.org/ontology/bio/>
@prefix meta: <http://bioportal.bioontology.org/ontology/meta/>
@prefix graphs: <http://bioportal.bioontology.org/ontology/graphs/>
@prefix omv: <http://bioportal.bioontology.org/ontology/omv/>

```

ONTOLOGY ID:	1353
BIOPORTAL PURL:	http://purl.bioontology.org/ontology/SNOMEDCT
STATUS:	
FORMAT:	UMLS
CATEGORIES:	Health
GROUPS:	Unified Medical Language System WHO Family of International Classifications Cancer Biomedical Informatics Grid
CONTACT:	Vivian A. Auld, auld@nlm.nih.gov
HOME PAGE:	http://ihtsdo.org
PUBLICATIONS PAGE:	http://ihtsdo.org
DOCUMENTATION PAGE:	http://ihtsdo.org
DESCRIPTION:	SNOMED Clinical Terms

```

<http://bioportal.bioontology.org/ontology/1353>
  meta:administeredBy "Vivian A. Auld"^^xsd:string ;
  meta:codingScheme "SNOMEDCT"^^xsd:string ;
  meta:fileNames "##/"^^xsd:string ;
  meta:filePath "/1353/"^^xsd:string ;
  meta:hasContactEmail "v.auld@nlm.nih.gov"^^xsd:string ;
  meta:hasContactName "Vivian A. Auld"^^xsd:string ;
  meta:hasDataGraph "http://purl.bioontology.org/ontology/SNOMEDCT"^^xsd:string ;
  meta:id 46896 ;
  meta:internalVersion "1.0"^^xsd:string ;
  meta:isFlat false ;
  meta:isFoundry false ;
  meta:isMetadataOnly false ;
  meta:isRemote false ;
  meta:isVersionOfVirtualOntology <http://bioportal.bioontology.org/ontologies/1353> ;
  meta:statusID 3 ;
  meta:targetTerminologies "SNOMEDCT"^^xsd:string ;
  meta:timestampCreation "2012-02-13T13:47:00"^^xsd:dateTime ;
  meta:urlHomepage "http://ihtsdo.org"^^xsd:string ;
  meta:urlPublications "http://ihtsdo.org"^^xsd:string ;
  omv:acronym "SNOMEDCT"^^xsd:string ;
  omv:creationDate "2011-07-31T00:00:00"^^xsd:dateTime ;
  omv:description "SNOMED Clinical Terms"^^xsd:string ;
  omv:documentation "http://ihtsdo.org"^^xsd:string ;

```

```

PREFIX xsd: <http://www.w3.org/2001/XMLSchema#>
PREFIX meta: <http://bioportal.bioontology.org/metadata/def/>
PREFIX omv: <http://omv.ontoware.org/2005/05/ontology#>

SELECT ?version ?name ?creation WHERE {
  ?version a omv:Ontology .
  ?version omv:name ?name .
  ?version meta:timestampCreation ?creation .
  FILTER (xsd:dateTime(?creation) > "2012-01-01T00:00:00"^^xsd:dateTime)
} ORDER BY ?name

```


version	name	creation
<http://bioportal.bioontology.org/ontologies/47573> ↗	"Adverse Event Reporting ontology"	"2012-05-31T18:06:33"
<http://bioportal.bioontology.org/ontologies/47302> ↗	"Adverse Event Reporting ontology"	"2012-04-25T18:06:40"
<http://bioportal.bioontology.org/ontologies/47035> ↗	"Adverse Event Reporting ontology"	"2012-03-09T18:08:24"
<http://bioportal.bioontology.org/ontologies/47022> ↗	"Adverse Event Reporting ontology"	"2012-03-07T18:07:48"
<http://bioportal.bioontology.org/ontologies/47058> ↗	"Adverse Event Reporting ontology"	"2012-03-13T18:10:57"
<http://bioportal.bioontology.org/ontologies/47432> ↗	"Adverse Event Reporting ontology"	"2012-05-11T18:06:38"
<http://bioportal.bioontology.org/ontologies/47523> ↗	"Adverse Event Reporting ontology"	"2012-05-23T18:06:28"
<http://bioportal.bioontology.org/ontologies/47581> ↗	"Anatomical Entity Ontology"	"2012-06-01T18:06:44"
<http://bioportal.bioontology.org/ontologies/46991> ↗	"Ascomycete phenotype ontology"	"2012-03-01T18:05:58"
<http://bioportal.bioontology.org/ontologies/47544> ↗	"BOOK"	"2012-05-27T19:19:21"
<http://bioportal.bioontology.org/ontologies/47507> ↗	"BRENDA tissue / enzyme source"	"2012-05-22T18:03:00"

All ontologies updated since *DATE*


```
PREFIX owl: <http://www.w3.org/2002/07/owl#>
PREFIX rdfs: <http://www.w3.org/2000/01/rdf-schema#>

SELECT DISTINCT ?graph WHERE {
  GRAPH ?graph {
    <http://mouse.brain-map.org/atlas/index.html#RHP> a owl:Class .
  }
}
```

graph

http://bioportal.bioontology.org/ontologies/H1_NMOABA_x2 

<http://bioportal.bioontology.org/ontologies/NMOABA> 

http://bioportal.bioontology.org/ontologies/H1_NMOABA 

http://bioportal.bioontology.org/ontologies/H1_NMOABA_x3 

http://bioportal.bioontology.org/ontologies/H1_NMOABA_x1 

<http://bioportal.bioontology.org/ontologies/ABA> 

Find term in all ontologies

```

PREFIX owl: <http://www.w3.org/2002/07/owl#>
PREFIX rdfs: <http://www.w3.org/2000/01/rdf-schema#>
PREFIX rdf: <http://www.w3.org/1999/02/22-rdf-syntax-ns#>

```

```

SELECT *
FROM <http://bioportal.bioontology.org/ontologies/ABA>
WHERE {
  ?s rdf:type owl:Class .
  ?s rdfs:label ?label .
}

```








s	label
http://mouse.brain-map.org/atlas/index.html#Alp6b	"Agranular insular area, posterior"
http://mouse.brain-map.org/atlas/index.html#BA	"Bed nucleus of the accessory olfa"
http://mouse.brain-map.org/atlas/index.html#NTSm	"Nucleus of the solitary tract, me"
http://mouse.brain-map.org/atlas/index.html#CA2	"Field CA2"
http://mouse.brain-map.org/atlas/index.html#LD	"Lateral dorsal nucleus of thalamu"
http://mouse.brain-map.org/atlas/index.html#SBPV	"Subparaventricular zone"
http://mouse.brain-map.org/atlas/index.html#SSp-ll5	"Primary somatosensory area, low"
http://mouse.brain-map.org/atlas/index.html#Alv6a	"Agranular insular area, ventral pa"
http://mouse.brain-map.org/atlas/in	
http://mouse.brain-map.org/atlas/in	
http://mouse.brain-map.org/atlas/in	

Select all terms from the ABA Adult Mouse Anatomy

```

PREFIX owl: <http://www.w3.org/2002/07/owl#>
PREFIX skos: <http://www.w3.org/2004/02/skos/core#>
SELECT DISTINCT ?termURI ?prefLabel
  FROM <http://bioportal.bioontology.org/ontologies/EHDA>
  FROM <http://bioportal.bioontology.org/ontologies/globals>
WHERE {
  ?termURI a owl:Class;
  skos:prefLabel ?prefLabel .
}

```

termURI	prefLabel
http://purl.obolibrary.org/obo/EHDA_6843 	"endodermal epithelium"
http://purl.obolibrary.org/obo/EHDA_6634 	"lateral wall"
http://purl.obolibrary.org/obo/EHDA_634 	"cavity"
http://purl.obolibrary.org/obo/EHDA_8119 	"genital tubercle"
http://purl.obolibrary.org/obo/EHDA_8816 	"medulla oblongata"
http://purl.obolibrary.org/obo/EHDA_6481 	"right part"
http://purl.obolibrary.org/obo/EHDA_6481 	

Select URI and preferred label from all terms


```

PREFIX owl: <http://www.w3.org/2002/07/owl#>
PREFIX rdfs: <http://www.w3.org/2000/01/rdf-schema#>

SELECT DISTINCT ?parent ?label WHERE {
GRAPH <http://bioportal.bioontology.org/ontologies/ABA> {
  <http://mouse.brain-map.org/atlas/index.html#RHP> rdfs:subClassOf ?parent .
  ?parent rdfs:label ?label .
}
}

```

parent

label









<http://mouse.brain-map.org/atlas/index.html#HPF>  "Hippocampal formation"

Preferred Name	Retrohippocampal region
ID	RHP
Full Id	http://mouse.brain-map.org/atlas/index.html#RHP
Is Defined By	http://mouse.brain-map.org/atlas/coronal/RHP.html
Label	Retrohippocampal region
Disjoint With	Hippocampal region
Sub Class Of	Hippocampal formation

Get parent of given term

```
PREFIX owl: <http://www.w3.org/2002/07/owl#>
PREFIX rdfs: <http://www.w3.org/2000/01/rdf-schema#>
PREFIX rdf: <http://www.w3.org/1999/02/22-rdf-syntax-ns#>
```












```
SELECT *
FROM <http://bioportal.bioontology.org/ontologies/BODYSYSTEM>
WHERE {
  ?s rdf:type owl:Class .
  ?s rdfs:label ?label .
  OPTIONAL { ?s rdfs:subClassOf ?parent . }
}
```

<http://who.int/bodysystem.owl#DigestiveSystem> 	"Digestive System"	<http://who.int/bodysys
<http://who.int/bodysystem.owl#AutonomicNervousSystem> 	"Autonomic Nervous System"	<http://who.int/bodysys
<http://who.int/bodysystem.owl#PeripheralNervousSystem> 	"Peripheral Nervous System"	<http://who.int/bodysys
<http://who.int/bodysystem.owl#BodySystem> 	"Body System"	---
<http://who.int/bodysystem.owl#VisualSystem> 	"Visual System"	<http://who.int/bodysys
<http://who.int/bodysystem.owl#UrinarySystem> 	"Urinary System"	<http://who.int/bodysys
<http://who.int/bodysystem.owl#FemaleGenitalSystem> 	"Female Genital System"	<http://who.int/bodysys
<http://who.int/bodysystem.owl#MetabolicSystem> 	"Metabolic"	<http://who.int/bodysys

Select all terms and their parent

```
SELECT DISTINCT ?p
FROM <http://bioportal.bioontology.org/ontologies/SNOMEDCT>
WHERE {
    [] ?p [] .
}
```


p

- <http://bioportal.bioontology.org/ontologies/umls/hasSTY> 
- <http://purl.bioontology.org/ontology/SNOMEDCT/focus_of> 
- <http://purl.bioontology.org/ontology/SNOMEDCT/procedure_context_of> 
- <http://purl.bioontology.org/ontology/SNOMEDCT/has_specimen_source_morphology> 
- <http://purl.bioontology.org/ontology/SNOMEDCT/has_scale_type> 
- <http://purl.bioontology.org/ontology/SNOMEDCT/indirect_device_of> 
- <http://purl.bioontology.org/ontology/SNOMEDCT/SUBSETMEMBER> 
- <http://purl.bioontology.org/ontology/SNOMEDCT/uses_access_device> 
- <http://purl.bioontology.org/ontology/SNOMEDCT/uses_energy> 
- <http://purl.bioontology.org/ontology/SNOMEDCT/has_associated_morphology> 
- <http://purl.bioontology.org/ontology/SNOMEDCT/has_intent> 


Select distinct properties

```
SELECT DISTINCT ?p WHERE {  
  GRAPH <http://bioportal.bioontology.org/ontologies/ABA> {  
    <http://mouse.brain-map.org/atlas/index.html#RHP> ?p ?o  
  }  
}
```

p

<<http://www.w3.org/2000/01/rdf-schema#isDefinedBy>> 

<<http://www.w3.org/2000/01/rdf-schema#subClassOf>> 

<<http://www.w3.org/1999/02/22-rdf-syntax-ns#type>> 

<<http://bioportal.bioontology.org/metadata/def/prefLabel>> 

<<http://www.w3.org/2002/07/owl#disjointWith>> 

<<http://www.w3.org/2000/01/rdf-schema#label>> 

Select properties for term

```
PREFIX owl: <http://www.w3.org/2002/07/owl#>
SELECT (COUNT(?term) as ?c)
FROM <http://bioportal.bioontology.org/ontologies/SNOMEDCT>
WHERE { ?term a owl:Class. }
```


c

"395036"

Metrics ?


NUMBER OF CLASSES:	395036
NUMBER OF INDIVIDUALS:	0
NUMBER OF PROPERTIES:	41
MAXIMUM DEPTH:	32
MAXIMUM NUMBER OF SIBLINGS:	10010
AVERAGE NUMBER OF SIBLINGS:	1
CLASSES WITH A SINGLE SUBCLASS:	<u>29002</u>
CLASSES WITH MORE THAN 25 SUBCLASSES:	<u>2053</u>
CLASSES WITH NO DEFINITION:	394352

Count terms in SNOMED



Performance Tips and Tricks

- Completely unbound patterns (?g ?s ?p ?o) are not allowed
- To optimize queries, use UNIONS instead of FILTERS
- If using FILTER on literals it is better if the filter is not applied to millions of rows
- To prevent combinatorial explosions of results, consider use CONSTRUCT or DESCRIBE (any M-N relationship can provoke this)




SPARQL Code Repository

- <https://github.com/ncbo/sparql-code-examples>

sparql-code-examples /

name	age	message
📁 java	4 months ago	Update java/src/org/ncbo/stanford/sparql/examples/SimpleTest.java [msalvadores]
📁 javascript	2 months ago	Simplified in-browser SPARQL call [Josh Mandel]
📁 perl	4 months ago	sparql.bioontology.org/sparql is now the production SPARQL endpoint [msalvadores]
📁 python	2 months ago	Eliminate references to alphasparql [Josh Mandel]
📁 ruby	4 months ago	Ruby examples for querying the BioPortal SPARQL endpoint [palexander]
📄 .gitignore	9 months ago	Javascript, Java, Perl, Python examples [msalvadores]
📄 COPYING	9 months ago	README and LICENSE [msalvadores]
📄 README	2 months ago	Eliminate references to alphasparql [Josh Mandel]



BioPortal SPARQL Endpoint

- Documentation:
http://www.bioontology.org/wiki/index.php/SPARQL_BioPortal
- Query interface: <http://sparql.bioontology.org/>
- Example queries:
<http://sparql.bioontology.org/examples>
- Sample code:
<https://github.com/ncbo/sparql-code-examples>